Shifted Partial Quadratics, their Pushforwards, and Signature Invariants for Tangles

Abstract. Following a general discussion of the computation of zombians of unfinished columbaria (with examples), I will tell you about my recent joint work w/ Jessica Liu on what we feel is the "textbook" extension of knot signatures to tangles, which for unknown reasons, is not in any of the textbooks that we know.


Jessica Liu



Columbaria in an East Sydney Cemetery


Image: Freepik.com

Jacobian, Hamiltonian, Zombian

Prior Art on signatures for tangles / braids. Gambaudo and Ghys [GG], Cimasoni and Conway [CC], Conway [Co], Merz [Me]. All define signatures of tangles / braids by first closing them to links and then work hard to derive composition properties.

Why Tangles? • Faster!

- Conceptually clearer proofs of invariance (and of skein relations).
- Often fun and consequential:
  - The Jones Polynomial ⇝ The Temperley-Lieb Algebra.
  - Khovanov Homology ⇝ "Unfinished complexes", complexes in a category.
  - The Kontsevich Integral ⇝ Associators.
  - HFK ⇝ OMG, type D, type A, $\mathcal{A}_\infty$, ...



$2^{n/2}+2^{n/2}+2^{\sqrt{n}} \ll 2^n$

Knot-tease


Zombies: Freepik.com

Computing Zombians of Unfinished Columbaria.

- Must be no slower than for finished ones.
- Future zombies must be able to complete the computation.
- Future zombies must not even know the size of the task that today's zombies were facing.
- We must be able to extend to ZPUCs, Zombie Processed Unfinished Columbaria!

Example / Exercise. Compute the determinant of a $1,000 \times 1,000$ matrix in which 50 entries are not yet given.


Columbarium near Assen

Homework / Research Projects. • What with ZPUCs? • Use this to get an Alexander tangle invariant.

Reminders. TL: {knots} → {matrices / quad. forms} $\xrightarrow[\sigma]{\text{signature}} \mathbb{Z}$:



With $|\omega|=1$, $t=1-\omega$, and $r=t+\bar t$, A is made by adding terms:



$$X_{-i,j,k,-l} \to \begin{array}{c}\begin{pmatrix} -r & -t & 2t & \bar t \\ -\bar t & 0 & \bar t & 0 \\ 2\bar t & t & -r & -\bar t \\ t & 0 & -t & 0 \end{pmatrix}\end{array}\begin{array}{c} i \\ j \\ k \\ l \end{array}$$



$$\bar X_{-i,j,k,-l} \to \begin{array}{c}\begin{pmatrix} r & -t & -2\bar t & \bar t \\ -\bar t & 0 & \bar t & 0 \\ -2t & t & r & -\bar t \\ t & 0 & -t & 0 \end{pmatrix}\end{array}\begin{array}{c} i \\ j \\ k \\ l \end{array}$$

On the occasion of my visit to Hungary I made a 63,202 Ft donation to the Háttér Society, `https://hatter.hu`.


Levine


Tristram

A quadratic form on a v.s. V over $\mathbb{C}$ is a quadratic $Q: V \to \mathbb{C}$, or a sesquilinear Hermitian $\langle\cdot,\cdot\rangle$ on $V\times V$ (so $\langle x,y\rangle = \overline{\langle y,x\rangle}$ and $Q(y)=\langle y,y\rangle$), or given a basis η_i of V^*, a matrix $A=(a_{ij})$ with $A=\bar A^T$ and $Q=\sum a_{ij}\bar\eta_i\eta_j$. The signature σ of Q is $\sigma_+-\sigma_-$, where for some P, $\bar P^T A P = \mathrm{diag}(1,\overset{\sigma_+}{\ldots},1,-1,\overset{\sigma_-}{\ldots},-1,0,\ldots)$.

A *Partial Quadratic (PQ)* on V is a quadratic Q defined only on a subspace $\mathcal{D}_Q \subset V$. We add PQs with $\mathcal{D}_{Q_1+Q_2} := \mathcal{D}_{Q_1}\cap\mathcal{D}_{Q_2}$. Given a linear $\psi: V\to W$ and a PQ Q on W, there is an obvious *pullback* ψ^*Q, a PQ on V.

Theorem 1. Given a linear $\phi: V\to W$ and a PQ Q on V, there is a unique *pushforward* PQ ϕ_*Q on W such that for every *PQ* U on W, $\sigma_V(Q+\phi^*U) = \sigma_{\ker\phi}(Q|_{\ker\phi}) + \sigma_W(U+\phi_*Q)$.

(If you must, $\mathcal{D}(\phi_*Q) = \phi(\mathrm{ann}_Q(\mathcal{D}(Q)\cap\ker\phi))$ and $(\phi_*Q)(w) = Q(v)$, where v is s.t. $\phi(v)=w$ and $Q(v, \mathrm{rad}\, Q|_{\ker\phi}) = 0$).

Gist of the Proof.



... and the quadratic $F =: \phi_*Q$ is well-defined only on $D := \ker C$.

Exactly what we want, if the Zombian is the signature!

V: The full space of *faces*.
W: The boundary, made of *gaps*.
Q: The known parts.
U: The part yet unknown.
$\sigma_V(Q+\phi^*(U))$: The overall Zombian.
$\sigma(Q|_{\ker\phi})$: An internal bit. $U+\phi_*Q$: A boundary bit.
And so our ZPUC is the pair $S=(\sigma(Q|_{\ker\phi}), \phi_*Q)$.



A *Shifted Partial Quadratic (SPQ)* on V is a pair $S=(s\in\mathbb{Z}, Q$ a PQ on $V)$. addition also adds the shifts, pullbacks keep the shifts, yet $\phi_*S := (s+\sigma_{\ker\phi}(Q|_{\ker\phi}), \phi_*Q)$ and $\sigma(S) := s+\sigma(Q)$.

Theorem 1' (*Reciprocity*). Given $\phi: V\to W$, for SPQs S on V and U on W we have $\sigma_V(S+\phi^*U) = \sigma_W(U+\phi_*S)$ (and this characterizes ϕ_*S).

Theorem 2. ψ^* and ϕ_* are functorial. Also, if $\alpha/\!\!/\beta = \gamma/\!\!/\delta$, α is surjective, β is injective, and $\mathrm{im}\,\gamma \supset \ker\delta$, then $\gamma^*/\!\!/\alpha_* = \delta_*/\!\!/\beta^*$. Finally, ψ^* is additive but ϕ_* isn't.

$$\begin{tikzcd} \bullet \arrow[r, "\alpha"] \arrow[d, "\gamma"'] & \bullet \arrow[d, "\beta"] \\ \bullet \arrow[r, "\delta"'] \arrow[ur, dashed] & \bullet \end{tikzcd}$$

Definition. $\mathcal{S}\left(\text{cyclic set } g_1, g_2, g_3, \ldots\right) := \left\{\begin{matrix}\text{SPQ } S \\ \text{on } \langle g_i\rangle\end{matrix}\right\}$.



Theorem 3. $\{\mathcal{S}(\text{cyclic sets})\}$ is a planar algebra, with compositions $\mathcal{S}(D)((S_i)) := \phi^D_*(\psi^*_D(\bigoplus_i S_i))$, where $\psi_D: \langle f_i\rangle \to \langle g_{\alpha i}\rangle$ maps every face of D to the sum of the input gaps adjacent to it and $\phi^D: \langle f_i\rangle \to \langle g_i\rangle$ maps every face to the sum of the output gaps adjacent to it. So for our D, ψ_D is $f_1\mapsto g_{34}$, $f_2\mapsto g_{31}+g_{14}+g_{24}+g_{33}$, $f_3\mapsto g_{32}$, $f_4\mapsto g_{11}$, $f_5\mapsto g_{13}+g_{21}$, $f_6\mapsto g_{23}$, $f_7\mapsto g_{12}+g_{22}$ and ϕ^D is $f_1\mapsto g_1$, $f_2\mapsto g_2+g_6$, $f_3\mapsto 0$, $f_4\mapsto g_3$, $f_5\mapsto 0$, $f_6\mapsto g_5$, $f_7\mapsto g_4$.



Connection Diagram

Theorem 4. TL, defined on X and $\bar X$ as before, extend to planar algebra morphisms TL: {tangles} → {$\mathcal{S}$}.

Comment. There's a nearly-identical "Kashaev version".


Kashaev

Implementation (sources: `http://drorbn.net/bu23/ap`). I like it most when the implementation matches the math perfectly. We failed here.

```
Once[<< KnotTheory`];
```

```
Loading KnotTheory` version
  of February 2, 2020, 10:53:45.2097.
Read more at http://katlas.org/wiki/KnotTheory.
```

Utilities. The step function, algebraic numbers, canonical forms.

```
θ[x_] /; NumericQ[x] := UnitStep[x]
```

```
ω2[v_][p_] := Module[{q = Expand[p], n, c},
  If[q === 0, 0,
   c = Coefficient[q, ω, n = Exponent[q, ω]];
   c v^n + ω2[v][q - c (ω + ω^-1)^n]]];
```

```
sign[ℰ_] := Module[{n, d, v, p, rs, e, k},
  {n, d} = NumeratorDenominator[ℰ];
  {n, d} /= ω^(Exponent[n,ω]/2+Exponent[n,ω,Min]/2);
  p = Factor[ω2[v]@n * ω2[v]@d /. v → 4 u^2 - 2];
  rs = Solve[p == 0, u, Reals];
  If[rs === {}, Sign[p /. u → 0],
   rs = Union@(u /. rs);
   Sign[(-1)^(e=Exponent[p,u]) Coefficient[p, u, e]] + Sum[
     k = 0;
     While[(d = RootReduce[∂_{u,++k} p /. u → r]) == 0];
     If[EvenQ[k], 0, 2 Sign[d]] * θ[u - r],
     {r, rs}]
   ]
  ]
```

```
SetAttributes[B, Orderless];
CF[b_B] := RotateLeft[#, First@Ordering[#] - 1] & /@
  DeleteCases[b, {}]
```

```
CF[ℰ_] := Module[{γs = Union@Cases[ℰ, γ_ | γ̄_, ∞]},
  Total[CoefficientRules[ℰ, γs] /.
    (ps_ → c_) ⧴ Factor[c] × Times @@ γs^ps] ]
```

```
CF[{}] = {};
CF[𝒞_List] :=
 Module[{γs = Union@Cases[𝒞, γ_, ∞], γ},
  CF /@ DeleteCases[0][
    RowReduce[Table[∂_γ r, {r, 𝒞}, {γ, γs}]].γs] ]
```

```
(ℰ_)* := ℰ /. {γ̄ → γ, γ → γ̄, ω → ω^-1, c_Complex ⧴ c*};
r_Rule* := {r, r*}
```

```
RulesOf[γ_i_ + rest_.] := (γ_i → -rest)*;
CF[PQ[𝒞_, q_]] := Module[{n𝒞 = CF[𝒞]},
  PQ[n𝒞, CF[q /. Union @@ RulesOf /@ n𝒞]] ]
```

```
CF[Σ_b_[σ_, pq_]] := Σ_CF[b][σ, CF[pq]]
```

Pretty-Printing.

```
Format[Σ_b_B[σ_, PQ[𝒞_, q_]]] := Module[{γs},
  γs = γ_# & /@ Join @@ b;
  Column[{TraditionalForm@σ,
    TableForm[Join[
      Prepend[""] /@ Table[TraditionalForm[∂_c r],
        {r, 𝒞}, {c, γs}],
      {Prepend[""][
        Join @@
         (b /. {l_, m___, r_} ⧴
            {DisplayForm@RowBox[{"(", l}],
             m, DisplayForm@RowBox[{r, ")"}]}) /.
         i_Integer ⧴ γ_i ]},
      MapThread[Prepend,
       {Table[TraditionalForm[∂_{r,c} q], {r, γs*},
         {c, γs}], γs*}]
      ], TableAlignments → Center]
   }, Center]];
```

The Face-Centric Core.

```
Σ_b1_[σ1_, PQ[𝒞1_, q1_]] ⊕ Σ_b2_[σ2_, PQ[𝒞2_, q2_]] ^:=
 CF@Σ_Join[b1,b2][σ1 + σ2, PQ[𝒞1 ⋃ 𝒞2, q1 + q2]];
```

GT for Gap Touch: 


```
GT_{i_,j_}@Σ_B[{li___,i_,ri___},{lj___,j_,rj___},bs___][σ_,
   PQ[𝒞_, q_]] :=
 CF@Σ_B[{ri,li,j,rj,lj,i},bs][σ, PQ[𝒞 ⋃ {γ_i - γ_j}, q]]
```



cor·don (kôr′dn)
n.
1. A line of people, military posts, or ships stationed around an area to enclose or guard it: *a police cordon.*
2. A rope, line, tape, or similar border stretched around an area, usually by the police, indicating that access is restricted.

$$\begin{array}{c} s \\ i \\ \end{array}\begin{pmatrix} 0 & \phi\, C_{\text{rest}} \\ \bar\phi^T & \lambda \quad \theta \\ \bar C^T_{\text{rest}} & \bar\theta^T A_{\text{rest}} \end{pmatrix} \to \begin{cases} \exists p\, \phi_p \neq 0 & \text{use } \phi_p \text{ to kill its row and column, drop a } \begin{pmatrix} 01 \\ 10 \end{pmatrix} \text{ summand} \\ \phi=0, \lambda\neq 0 & \text{use } \lambda \text{ to kill } \theta\text{, let } s \mathrel{+}= \text{sign}(\lambda) \\ \phi=0, \lambda=0 & \text{append } \theta \text{ to } C_{\text{rest}}. \end{cases}$$

```
Cordon_i_@Σ_B[{li___,i_,ri___},bs___][σ_, PQ[𝒞_, q_]] :=
 Module[{ϕ = ∂_{γ_i} 𝒞, λ = ∂_{γ̄_i,γ_i} q, nσ = σ, n𝒞, nq, p},
  {p} = FirstPosition[(# =!= 0) & /@ ϕ, True, {0}];
  {n𝒞, nq} = Which[
    p > 0, {𝒞, q} /. (γ_i → -𝒞[[p]] / ϕ[[p]])* /. (γ_i → 0)*,
    λ =!= 0, (nσ += sign[λ];
     {𝒞, q /. (γ_i → -(∂_{γ̄_i} q) / λ)* /. (γ_i → 0)*}),
    λ === 0, {𝒞 ⋃ {∂_{γ̄_i} q}, q /. (γ_i → 0)*}];
  CF@Σ_B[Most@{ri,li},bs][nσ,
    PQ[n𝒞, nq] /. (γ_Last@{ri,li} → γ_First@{ri,li})*] ]
```

Strand Operations. c for contract, mc for magnetic contract:

```
c_{i_,j_}@t : Σ_{B[{li___,i_,ri___},{___,j_,___},___]}[__] :=
  t // GT_{j,First@{ri,li}} // Cordon_j
c_{i_,j_}@t : Σ_{B[{___,i_,j_,___},___]}[__] := Cordon_j@t
c_{i_,j_}@t : Σ_{B[{j_,___,i_},___]}[__] := Cordon_j@t
c_{i_,j_}@t : Σ_{B[{___,j_,i_,___},___]}[__] := Cordon_i@t
c_{i_,j_}@t : Σ_{B[{i_,___,j_},___]}[__] := Cordon_i@t
mc[ℰ_] := ℰ //.
  t : Σ_{B[{___,i_,___},{___,j_,___},___]}[__] |
     Σ_{B[{___,i_,j_,___},___]}[__] | Σ_{B[{j_,___,i_},___]}[__] /;
    i + j == 0 :> c_{i,j}@t
```

The Crossings (and empty strands).

```
TL@P_{i_,j_} := CF@Σ_{B[{i,j}]}[0, PQ[{}, 0]]
TL[x : X[i_, j_, k_, l_]] :=
  TL@If[PositiveQ[x], X_{-i,j,k,-l}, X̄_{-j,k,l,-i}];
TL[(x : X | X̄)_{fs__}] := Module[{t = 1 - ω, r, γs, m},
  r = t + t*; γs = γ_# & /@ {fs};
  m = If[x === X,
     ( -r   -t   2t   t*
       -t*  0    t*   0
       2t*  t    -r   -t*
       t    0    -t   0 ),
     ( r    -t   -2t*  t*
       -t*  0    t*    0
       -2t  t    r     -t*
       t    0    -t    0 )];
  CF@Σ_{B[{fs}]}[0, PQ[{}, γs*.m.γs]]]
```

Evaluation on Tangles and Knots.

```
TL[K_] :=
  Fold[mc[#1 ⊕ #2] &, Σ_{B[]}[0, PQ[{}, 0]],
    List @@ (TL /@ PD@K)] /.
   θ[c_ + u] /; Abs[c] ≥ 1 :> θ[c];
TLSig[K_] := TL[K][[1]]
```

Reidemeister 3.




```
R3L = PD[X_{-2,5,4,-1}, X_{-3,7,6,-5},
   X_{-6,9,8,-4}];
R3R = PD[X_{-3,5,4,-2}, X_{-4,6,8,-1},
   X_{-5,7,9,-6}];
TL@R3L == TL@R3R
```

True

```
TL@R3L
```

$$-1$$

$$\begin{array}{c|cccccc} & \gamma_{-3} & \gamma_7 & \gamma_9 & \gamma_8 & \gamma_{-1} & \gamma_{-2} \\ \hline \bar\gamma_{-3} & \frac{\omega^2+1}{\omega} & \omega-1 & -2\omega & 2 & 0 & -\frac{\omega+1}{\omega} \\ \bar\gamma_7 & -\frac{\omega-1}{\omega} & 0 & \frac{\omega-1}{\omega} & 0 & 0 & 0 \\ \bar\gamma_9 & -\frac{2}{\omega} & 1-\omega & \frac{\omega^2+1}{\omega} & -\frac{\omega+1}{\omega} & 0 & \frac{2}{\omega} \\ \bar\gamma_8 & 2 & 0 & -\omega-1 & \frac{\omega^2+1}{\omega} & -\frac{\omega-1}{\omega} & -\frac{2}{\omega} \\ \bar\gamma_{-1} & 0 & 0 & 0 & \omega-1 & 0 & 1-\omega \\ \bar\gamma_{-2} & -\omega-1 & 0 & 2\omega & -2\omega & \frac{\omega-1}{\omega} & \frac{\omega^2+1}{\omega} \end{array}$$

Reidemeister 2.




```
TL@PD[X̄_{-2,4,3,-1}, X̄_{-4,6,5,-3}]
```

$$0$$

$$\begin{array}{c|cccc} & 1 & 0 & -1 & 0 \\ & \gamma_{-2} & \gamma_6 & \gamma_5 & \gamma_{-1} \\ \hline \bar\gamma_{-2} & 0 & 0 & 0 & 0 \\ \bar\gamma_6 & 0 & 0 & 0 & 0 \\ \bar\gamma_5 & 0 & 0 & 0 & 0 \\ \bar\gamma_{-1} & 0 & 0 & 0 & 0 \end{array}$$

```
TL@PD[X̄_{-2,4,3,-1}, X̄_{-4,6,5,-3}] == GT_{5,-2}@TL@PD[P_{-1,5}, P_{-2,6}]
```

True

Reidemeister 1.




```
TL@PD[X_{-3,3,2,-1}] == TL@P_{-1,2}
```

True

A Knot.



```
f = TLSig[Knot[8, 5]]
```

$$2\,\theta\left[-\frac{\sqrt{3}}{2}+u\right]-2\,\theta\left[\frac{\sqrt{3}}{2}+u\right]-2\,\theta\left[u-(-0.630\ldots)\right]+2\,\theta\left[u-0.630\ldots\right]$$

```
Plot[f, {u, -1, 1}]
```



The Conway-Kinoshita-Terasaka Tangles.



Conway Kinoshita Terasaka






```
T1 = PD[X̄_{-6,2,7,-1}, X̄_{-2,8,3,-7},
   X̄_{-8,4,9,-3}, X_{-11,6,12,-5},
   X_{-4,11,5,-10}];
T2 = PD[X_{-6,2,7,-1}, X_{-2,8,3,-7},
   X_{-8,4,9,-3}, X̄_{-12,6,13,-5},
   X̄_{-4,12,5,-11}, X̄_{-10,15,11,-14}, X̄_{-15,10,16,-9}];
TL[T1]
```

$$-2\,\theta\left(u-\frac{\sqrt{3}}{2}\right)+2\,\theta\left(u+\frac{\sqrt{3}}{2}\right)-1$$

$$\begin{array}{c|cccc} & \gamma_{-10} & \gamma_9 & \gamma_{-1} & \gamma_{12} \\ \hline \bar\gamma_{-10} & 0 & 1-\omega & 0 & \omega-1 \\ \bar\gamma_9 & \frac{\omega-1}{\omega} & \frac{2\omega}{\omega^2-\omega+1} & -\frac{\omega-1}{\omega} & -\frac{2\omega}{\omega^2-\omega+1} \\ \bar\gamma_{-1} & 0 & \omega-1 & 0 & 1-\omega \\ \bar\gamma_{12} & -\frac{\omega-1}{\omega} & -\frac{2\omega}{\omega^2-\omega+1} & \frac{\omega-1}{\omega} & \frac{2\omega}{\omega^2-\omega+1} \end{array}$$

```
TL[T2]
```

$$0$$

$$\begin{array}{c|cccc} & \gamma_{-14} & \gamma_{16} & \gamma_{-1} & \gamma_{13} \\ \hline \bar\gamma_{-14} & 0 & 1-\omega & 0 & \omega-1 \\ \bar\gamma_{16} & \frac{\omega-1}{\omega} & -\frac{2(\omega-1)^2\omega}{\omega^4-3\omega^3+5\omega^2-3\omega+1} & -\frac{\omega-1}{\omega} & \frac{2(\omega-1)^2\omega}{\omega^4-3\omega^3+5\omega^2-3\omega+1} \\ \bar\gamma_{-1} & 0 & \omega-1 & 0 & 1-\omega \\ \bar\gamma_{13} & -\frac{\omega-1}{\omega} & \frac{2(\omega-1)^2\omega}{\omega^4-3\omega^3+5\omega^2-3\omega+1} & \frac{\omega-1}{\omega} & -\frac{2(\omega-1)^2\omega}{\omega^4-3\omega^3+5\omega^2-3\omega+1} \end{array}$$

Examples with non-trivial co-dimension.




```
B1 = PD[X_{-5,2,6,-1}, X̄_{-8,3,9,-2},
   X_{-11,4,12,-3}, X_{-12,10,13,-9},
   X̄_{-13,7,14,-6}];
B2 = PD[X_{-5,2,6,-1}, X̄_{-9,3,10,-2},
   X_{-10,7,11,-6}, X̄_{-12,4,13,-3}, X_{-13,8,14,-7}];
TL[B1]
```

0

	1	0	-1	0	$\frac{1}{\omega}$	0	$-\frac{1}{\omega}$	0
	0	0	0	-1	$\frac{1}{\omega}$	0	$-\frac{1}{\omega}$	1
	$(\gamma_{-11}$	γ_4	γ_{10}	γ_7	γ_{14}	γ_{-1}	γ_{-5}	$\gamma_{-8})$
$\bar\gamma_{-11}$	0	0	0	0	0	0	0	0
$\bar\gamma_4$	0	0	0	0	$\frac{\omega-1}{\omega^2}$	0	$-\frac{\omega-1}{\omega^2}$	0
$\bar\gamma_{10}$	0	0	0	0	$-\frac{\omega-1}{\omega}$	0	$\frac{\omega-1}{\omega}$	0
$\bar\gamma_7$	0	0	0	0	$\frac{(\omega-1)^2}{\omega^2}$	0	$-\frac{(\omega-1)^2}{\omega^2}$	0
$\bar\gamma_{14}$	0	$-((\omega-1)\ \omega)$	$\omega-1$	$(\omega-1)^2$	0	$-\frac{\omega-1}{\omega}$	$\frac{\omega-1}{\omega}$	0
$\bar\gamma_{-1}$	0	0	0	0	$\omega-1$	0	$1-\omega$	0
$\bar\gamma_{-5}$	0	$(\omega-1)\ \omega$	$1-\omega$	$-(\omega-1)^2$	$1-\omega$	$\frac{\omega-1}{\omega}$	$\frac{(\omega-1)^2}{\omega}$	0
$\bar\gamma_{-8}$	0	0	0	0	0	0	0	0

```
TL[B2]
```

0

	$(\gamma_{-12}$	γ_4	γ_8	γ_{14}	γ_{11}	γ_{-1}	γ_{-5}	$\gamma_{-9})$
$\bar\gamma_{-12}$	$\frac{(\omega-1)^2}{\omega}$	$\omega-1$	$-2(\omega-1)$	$\frac{2(\omega-1)^2}{\omega}$	$\frac{2(\omega-1)}{\omega^2}$	0	$-\frac{2(\omega-1)}{\omega^2}$	$-\frac{(\omega-1)(2\omega-3)}{\omega}$
$\bar\gamma_4$	$-\frac{\omega-1}{\omega}$	0	$\frac{\omega-1}{\omega}$	0	0	0	0	0
$\bar\gamma_8$	$\frac{2(\omega-1)}{\omega}$	$1-\omega$	$\frac{(\omega-1)^2}{\omega}$	$-\frac{(\omega-1)(2\omega-3)}{\omega}$	$-\frac{2(\omega-1)}{\omega^2}$	0	$\frac{2(\omega-1)}{\omega^2}$	$\frac{2(\omega-2)(\omega-1)}{\omega}$
$\bar\gamma_{14}$	$\frac{2(\omega-1)^2}{\omega}$	0	$-\frac{(\omega-1)(3\omega-2)}{\omega}$	$\frac{3(\omega-1)^2}{\omega}$	$-\frac{(\omega-2)(\omega-1)}{\omega^2}$	0	$-\frac{2(\omega-1)}{\omega^2}$	$-\frac{2(\omega-2)(\omega-1)}{\omega}$
$\bar\gamma_{11}$	$-2(\omega-1)\ \omega$	0	$2(\omega-1)\ \omega$	$-((\omega-1)(2\omega-1))$	$\frac{(\omega-1)^2}{\omega}$	$-\frac{\omega-1}{\omega}$	$\frac{2(\omega-1)}{\omega}$	$2(\omega-1)^2$
$\bar\gamma_{-1}$	0	0	0	0	$\omega-1$	0	$1-\omega$	0
$\bar\gamma_{-5}$	$2(\omega-1)\ \omega$	0	$-2(\omega-1)\ \omega$	$2(\omega-1)\ \omega$	$-2(\omega-1)$	$\frac{\omega-1}{\omega}$	$\frac{(\omega-1)^2}{\omega}$	$-((\omega-1)(2\omega-1))$
$\bar\gamma_{-9}$	$-\frac{(\omega-1)(3\omega-2)}{\omega}$	0	$\frac{2(\omega-1)(2\omega-1)}{\omega}$	$-\frac{2(\omega-1)(2\omega-1)}{\omega}$	$\frac{2(\omega-1)^2}{\omega^2}$	0	$-\frac{(\omega-2)(\omega-1)}{\omega^2}$	$\frac{3(\omega-1)^2}{\omega}$

$$\begin{pmatrix}A & B\\ C & U\end{pmatrix} \xrightarrow{\det(A)} \begin{pmatrix}I & A^{-1}B\\ C & U\end{pmatrix} \xrightarrow{1} \begin{pmatrix}I & A^{-1}B\\ 0 & U-CA^{-1}B\end{pmatrix},$$

so $\det\begin{pmatrix}A & B\\ C & U\end{pmatrix} = \det(A)\det(U-CA^{-1}B)$. (what if $\nexists A^{-1}$?)

Questions. 1. Does this have a topological meaning? 2. Is there a "Kashaev conjecture" for tangles? 3. Find all solutions of R123 in our "algebra". 4. Braids and the Burau representation. 5. Recover the work in "Prior Art". 6. Are there any concordance properties? 7. What is the "SPQ group"? 8. The jumping points of signatures are the roots of the Alexander polynomial. Does this generalize to tangles? 9. Which of the three Cordon cases is the most common? 10. Are there interesting examples of tangles for which rels is non-trivial? 11. Is the pq part determined by Γ-calculus? 12. Is the pq part determined by finite type invariants? 13. Does it work with closed components / links? 14. Strand-doubling formulas? 15. A multivariable version? 16. Mutation invariance? 17. Ribbon knots? 18. Are there "face-virtual knots"? 19. Does the pushforward story extend to ranks? To formal Gaussian measures? To super Gaussian measures?

Proof of Theorem 1.

Uniqueness: If A and B are 2 pushforwards, then $\sigma_W(U+A) = \sigma_W(U+B)$ for all PQs U on W.
Thus $\mathcal{D}_A = \mathcal{D}_B$, because otherwise if $w \in \mathcal{D}_A \setminus \mathcal{D}_B$, by taking $U(w) = 1$ on $\mathcal{D}_U = \operatorname{span}\{w\}$, we get $\sigma_W(U+A) = 1 \neq 0 = \sigma_W(U+B)$. Furthermore, A and B must agree where they are both defined, because by taking $U(w) = \frac{-A(w)-B(w)}{2}$ on $\mathcal{D}_U = \operatorname{span}\{w\}$ we get $(U+A)(w) = \frac{A(w)-B(w)}{2} = -(U+B)(w)$, so we must have $A(w) = B(w)$ to satisfy $\sigma_W(U+A) = \sigma_W(U+B)$.

Existence: Define ϕ_*Q by $\mathcal{D}_{\phi_*Q} = \phi(\operatorname{ann}_Q(\ker\phi))$ and $\phi_*Q(w) = Q(v)$ where $v \in \operatorname{ann}_Q(\ker\phi)$. Note that ϕ_*Q is well-defined.
First consider when $U = 0$ on all of W. Let K be a maximal non-degenerate subspace of $\ker\phi$. Then $Q = Q|_K \oplus Q|_{\operatorname{ann}_Q(K)}$, and we can write $\operatorname{ann}_Q(K) = R \oplus A \oplus B$ where $R = \operatorname{rad}_Q(\ker\phi)$ and A, B are chosen so that $A \subseteq \operatorname{ann}_Q(R)$ and $B \subseteq \operatorname{ann}_Q(K) \setminus \operatorname{ann}_Q(R)$. Since $Q : R \to B^*$ is surjective, for any $v \in \mathcal{D}_Q$ there is some $r_v \in R$ such that $Q(r_v, B) = Q(v, B)$. If we choose the r_v so that $r_{v_1} + r_{v_2} = r_{v_1+v_2}$, then we can replace A by $A' = \{a - r_a : a \in A\}$ and B by $B' = \{b - \frac{1}{2}r_b : b \in B\}$ to get $Q = Q|_K \oplus Q|_{R\oplus B'} \oplus Q|_{A'}$.
Then notice that

- $\sigma_V(Q|_K) = \sigma_{\ker\phi}(Q|_{\ker\phi})$
- $\sigma_V(Q|_{R\oplus B'}) = 0$
- $\sigma_V(Q|_{A'}) = \sigma_W(\phi_*Q)$

so we get $\sigma_V(Q) = \sigma_{\ker\phi}(Q|_{\ker\phi}) + \sigma_W(\phi_*Q)$.
Now for an arbitrary U, note that $(Q+\phi^*U)|_{\ker\phi} = Q|_{\ker\phi}$ and $\phi_*(Q+\phi^*U) = \phi_*Q + U$ so we can replace Q in the $U = 0$ case by $Q + \phi^*U$ to get the general case.

Proof of Theorem 2.

It's clear that pullback is functorial and that pushforward by the identity is the identity. To show $(\phi\psi)_* = \phi_*\psi_*$, use theorem 1 repeatedly to get

$$\begin{aligned}
&\sigma((\phi\psi)_*Q + U)\\
=&\sigma(Q + (\phi\psi)^*U)\\
=&\sigma(Q + \psi^*\phi^*U) - \sigma(Q|_{\ker\phi\psi})\\
=&\sigma(\psi_*Q + \phi^*U) + \sigma(Q|_{\ker\psi}) - \sigma(Q|_{\ker\phi\psi})\\
=&\sigma(\phi_*\psi_*Q + U) + \sigma(Q|_{\ker\psi}) + \sigma(\psi_*Q|_{\ker\phi}) - \sigma(Q|_{\ker\phi\psi})\\
=&\sigma(\phi_*\psi_*Q + U)
\end{aligned}$$

for any U, where the last step uses theorem 1 on $Q|_{\phi\psi}$ with the map $\psi : \ker\phi\psi \to \ker\phi$.

$$\begin{tikzcd} \bullet \arrow[r, "\alpha"] \arrow[d, "\gamma"'] & \bullet \arrow[d, "\beta"] \\ \bullet \arrow[r, "\delta"'] \arrow[ur, dotted] & \bullet \end{tikzcd}$$

To show $\alpha_*\gamma^* = \beta^*\delta_*$, first note that $\beta^*\beta_*$ is the identity on any PQ since β is injective, so

$$\alpha_*\gamma^*Q = \beta^*(\beta\alpha)_*\gamma^*Q = \beta^*(\delta\gamma)_*\gamma^*Q = \beta^*\delta_*\gamma_*\gamma^*Q$$

As $\beta^*\delta_*\gamma_*\gamma^*Q$ and $\beta^*\delta_*Q$ have the same values where they are both defined, it remains to show that they have the same domain. Since α is surjective and γ is surjective onto $\ker(\delta)$, we see that

$$\beta^{-1}\delta(A) = \beta^{-1}\delta(A \cap \operatorname{im}\gamma)$$

for any subspace A. By taking $A = \operatorname{ann}_Q(\ker\delta)$, the two sides of the equality become the domains of $\beta^*\delta_*Q$ and $\beta^*\delta_*\gamma_*\gamma^*Q$.

References.

[CC] D. Cimasoni, A. Conway, *Colored Tangles and Signatures,* Math. Proc. Camb. Phil. Soc. **164** (2018) 493–530, arXiv:1507.07818.

[Co] A. Conway, *The Levine-Tristram Signature: A Survey,* arXiv:1903.04477.

[GG] J-M. Gambaudo, É. Ghys, *Braids and Signatures,* Bull. Soc. Math. France **133-4** (2005) 541–579.

[Ka] R. Kashaev, *On Symmetric Matrices Associated with Oriented Link Diagrams,* in *Topology and Geometry, A Collection of Essays Dedicated to Vladimir G. Turaev,* EMS Press 2021, arXiv:1801.04632.

[Me] A. Merz, *An Extension of a Theorem by Cimasoni and Conway,* arXiv:2104.02993.

Acknowledgement. This work was partially supported by NSERC grant RGPIN-2018-04350 and by the Chu Family Foundation (NYC).